Weekly Assignment #5

Instructions: **integrations in problems 1 and 2 should be evaluated by hand. <u>Some</u> numerical integrations should also be done by hand (whenever n ≤ 4); if n>4, you may use technology. Show your work!**

1. Average value

Let $f(x) = x(e^{2x} + e^{3x})$.

a. Define and plot $f(x)$ on the interval [0, 1/2].

In[2479]:=
```
f[x_] := x (Exp[2 x] + Exp[3 x])
a = 0;
b = 1/2;
Plot[f[x], {x, a, b}]
```

Out[2482]=



2. Demonstrate integration by parts to find the true average value of $f(x)$ on [0, 1/2]. Give the name "true" to this average value.

In[2483]:=
```
true = 1/(b - a) Integrate[f[x], {x, a, b}]
N[true]
```

Out[2483]= $\frac{1}{18}\left(13 + 2\, e^{3/2}\right)$

Out[2484]= 1.22019

3. Use the Trapezoidal and Midpoint Rules with n=2 to approximate the average value over the interval [0,1/2] (calls these values "trap" and "mid"). Combine them in an appropriate way to give the Simpson's Rule ("simp") approximation, with n=4 (note: this n=4 value S_4 uses a weighted average of $T_2$and M_2).

In[2485]:=
```
n = 2;
deltaX = (b - a) / n;
trap = 1.0 / (b - a) * deltaX / 2 * (f[a] + 2 f[a + deltaX] + f[b])
mid = 1.0 / (b - a) * deltaX * (f[a + deltaX / 2] + f[a + 3 / 2 deltaX])
simp = (2 * mid + trap) / 3
```

Out[2487]= 1.37071

Out[2488]= 1.14567

Out[2489]= 1.22068

4. Compute the errors (true - estimate) for the Trapezoidal and Midpoint rules. By plotting and bounding the absolute value of the second derivative of f(x), demonstrate that the approximations are within the error bounds for each method. Demonstrate that the errors are of opposite sign, as is typical.

In[2490]:=
```
Plot[Abs[f''[x]], {x, a, b}]
K = N[f''[b]] / (b - a) (* looks like we can take the right endpoint value for K;
notice that I divide by (b-a), since that's actually the integral that is
 being approximated! The integrand can be considered f(x)/(b-a).   *)
trapBound = K (b - a) ^3 / (12 n^2);
trapError = true - trap;
{trapError, trapBound}
midBound = K (b - a) ^3 / (24 n^2);
midError = true - mid;
{midError, midBound}
```

Out[2490]=


Out[2491]= 126.735

Out[2494]= {-0.150524, 0.330039}

Out[2497]= {0.074521, 0.165019}

5. Compute the absolute error (|true - estimate|) for the Simpson's Rule approximation. By plotting the absolute value of the fourth derivative of f(x) (choose K as big as, or bigger than, the largest value), demonstrate that the approximations are within the error bounds for Simpson's Rule.

In[2498]:=
```
Plot[Abs[f''''[x]], {x, a, b}]
K = N[f''''[b]] / (b - a) (* Looks like it's greatest on the right endpoint *)
bound = K (b - a) ^5 / (180 (2 n) ^4);
simpError = true - simp;
{simpError, bound}
```

Out[2498]=



Out[2499]= 1548.52

Out[2502]= {-0.000493976, 0.00105016}

2. Area and Volume

Let R be the region in the first quadrant of the plane between the curve $y = f(x) = \sqrt{4 - x^2}$ and the x-axis.

1. Define and plot $f(x)$ on the interval [0, 2].

In[2503]:=
```
f[x_] := Sqrt[4 - x^2]
a = 0;
b = 2;
Plot[f[x], {x, 0, 2}]
```

Out[2506]=



2. Find the area of R, by computing the appropriate integral using trigonometric substitution.

In[2507]:=
```
true = Integrate[f[x], {x, a, b}]
```
Out[2507]= π

3. Estimate the area using S_{10}.

In[2508]:=
```
trapezoidal[exp_, {x_, a_, b_, n_}] := Module[{dx = N[(b - a) / n], f},
   f[u_] := exp /. x → u;
   Return[0.5 dx (f[a] + f[b] + 2 Sum[f[a + k dx], {k, 1, n - 1}])]
  ]
midpoint[exp_, {x_, a_, b_, n_}] := Module[{dx = N[(b - a) / n], f},
   f[u_] := exp /. x → u;
   Return[dx Sum[f[a + k dx - 1 / 2 dx], {k, 1, n}]]
  ]
simpson[exp_, {x_, a_, b_, n_}] := If[EvenQ[n],
   (trapezoidal[exp, {x, a, b, n / 2}] + 2 midpoint[exp, {x, a, b, n / 2}]) / 3,
   Print["Sorry: Simpson's Rule only works on an even number of sub-intervals."]
  ]
simp = simpson[f[x], {x, a, b, 10}]
```
Out[2511]= 3.12701

4. Compute the absolute error in the approximation.

In[2512]:=
```
simpError = Abs[true - simp]
```
Out[2512]= 0.0145845

5. Compute the fourth derivative of *f(x)*. Why can we not use the error estimate for Simpson's rule in this case? (Is it possible to bound the fourth derivative on this interval?)

In[2513]:=
```
Simplify[f''''[x]]
Plot[Abs[f''''[x]], {x, a, b}]
```
Out[2513]= $-\frac{48\left(1+x^2\right)}{\left(4-x^2\right)^{7/2}}$

Out[2514]=


3. Application

When we die, our bodies become rigid (*rigor mortis* sets in). Niderkorn's (1872) observations on 113 bodies provides the main reference database for the development of *rigor mortis*.

The data:

hour	2	3	4	5	6	7	8	9	10	11	12	1
proportion in *rigor mortis*	$\frac{2}{123}$	$\frac{16}{123}$	$\frac{47}{123}$	$\frac{61}{123}$	$\frac{81}{123}$	$\frac{92}{123}$	$\frac{99}{123}$	$\frac{103}{123}$	$\frac{110}{123}$	$\frac{111}{123}$	$\frac{111}{123}$	$\frac{11}{12}$



One can fit a lovely model to this somewhat unlovely data, for the proportion *p(t)* of bodies in complete *rigor mortis* after *t* hours. It is illustrated in the graph below:



The model is $e^{-\frac{22.47}{t^{2.216}}}$: that is,

In[2521]:=
```
p[t_] := E^(-22.47 t^(-2.216))
```

a. Compute the average proportion of bodies in rigor mortis in the time interval from 6 to 10 hours after death, based on this model (write the integral, but you may use your calculator or Mathematica to produce your answer!).

```
NIntegrate[p[t] / (10 - 6), {t, 6, 10}]
(* So the average proportion of bodies you'd expect to see in the 6-
 10 hour mark passed the time of death is about 79%. *)
```

Out[2522]= 0.7876

b. Now compute an approximation using only the data, and Simpson's rule with n=4 (by hand). How do they compare?

In[2528]:=
```
a = 6;
b = 10;
n = 4;
deltaX = (b - a) / n;
simp = 1 / (b - a) (deltaX / 3.0 (81 + 4 * 92 + 2 * 99 + 4 * 103 + 110) / 123)
```

Out[2532]= 0.792005